Math 132

Summer 2009 Exam II

1. Calculate $\frac{d}{dx} 3^{\ln(x)}$.

```
> general_formula := Diff(3^u,u) = diff(3^u,u);
```

$$general\_formula := \frac{d}{du} 3^u = 3^u \ln(3)$$

```
> Chain_Rule_form := Diff(3^u(x),x) = diff(3^u(x),x);
```

$$Chain\_Rule\_form := \frac{d}{dx} 3^{u(x)} = 3^{u(x)} \left(\frac{d}{dx} u(x)\right) \ln(3)$$

```
> given_problem := subs(u(x) = ln(x), Chain_Rule_form);
```

$$given\_problem := \frac{d}{dx} 3^{\ln(x)} = 3^{\ln(x)} \left(\frac{d}{dx} \ln(x)\right) \ln(3)$$

```
> answer := lhs(given_problem) = rhs(given_problem);
```

$$answer := \frac{d}{dx} 3^{\ln(x)} = \frac{3^{\ln(x)} \ln(3)}{x}$$

2. Calculate $\int_1^2 \log_2(x)\, dx$.

```
> with(student):
> indefinite_integral := Int(log[2](x),x);
```

$$indefinite\_integral := \int \frac{\ln(x)}{\ln(2)}\, dx$$

```
> after_integration_by_parts :=
  intparts(indefinite_integral,ln(x));
```

$$after\_integration\_by\_parts := \frac{\ln(x)\, x}{\ln(2)} - \int \frac{1}{\ln(2)}\, dx$$

```
> antiderivative := value(after_integration_by_parts);
```

$$antiderivative := \frac{\ln(x)\,x}{\ln(2)} - \frac{x}{\ln(2)}$$

```
> answer := subs(x = 2, antiderivative) - subs(x = 1,
  antiderivative);
```

$$answer := 2 - \frac{1}{\ln(2)} - \frac{\ln(1)}{\ln(2)}$$

```
> answer := simplify(answer);
```

$$answer := \frac{2\ln(2) - 1}{\ln(2)}$$

3. Calculate $\frac{d}{dx}\, x^{\left(\frac{1}{x}\right)}$.

```
> eqn := f(x) = x^(1/x);
```

$$eqn := f(x) = x^{\left(\frac{1}{x}\right)}$$

```
> eqn2 := map(ln, eqn);
```

$$eqn2 := \ln(f(x)) = \ln\left(x^{\left(\frac{1}{x}\right)}\right)$$

```
> eqn3 := map(expand, eqn2) assuming x :: positive;
```

$$eqn3 := \ln(f(x)) = \frac{\ln(x)}{x}$$

```
> eqn4 := map( z -> diff(z, x), eqn3);
```

$$eqn4 := \frac{\frac{d}{dx} f(x)}{f(x)} = -\frac{\ln(x)}{x^2} + \frac{1}{x^2}$$

```
> eqn5 := Diff(f(x), x) = f(x)*rhs(eqn4);
```

$$eqn5 := \frac{d}{dx} f(x) = f(x)\left(-\frac{\ln(x)}{x^2} + \frac{1}{x^2}\right)$$

```
> Diff(f(x), x) = subs(eqn, rhs(eqn5));
```

$$\frac{d}{dx} f(x) = x^{\left(\frac{1}{x}\right)}\left(-\frac{\ln(x)}{x^2} + \frac{1}{x^2}\right)$$

4. A radioactive substance has mass 120g at time $t = 2$ and mass 80g at time $t = 4$. What is the mass at $t = 10$?

```
> restart;  interface(showassumed = 0): assume(k,positive);
> basic_eqn := m = A*exp(-k*t);
```

$$basic\_eqn := m = A\,\mathbf{e}^{(-k\,t)}$$

```
> eqn1 := subs({m = 120, t = 2}, basic_eqn);
```

$$eqn1 := 120 = A\,\mathbf{e}^{(-2\,k)}$$

```
> eqn2 := subs({m = 80, t = 4}, basic_eqn);
```

$$eqn2 := 80 = A\,\mathbf{e}^{(-4\,k)}$$

```
> eqn3 := lhs(eqn1)/lhs(eqn2) = rhs(eqn1)/rhs(eqn2);
```

$$eqn3 := \frac{3}{2} = \frac{\mathbf{e}^{(-2\,k)}}{\mathbf{e}^{(-4\,k)}}$$

```
> eqn4 := ln(lhs(eqn3)) = simplify(ln(rhs(eqn3)));
```

$$eqn4 := \ln\left(\frac{3}{2}\right) = 2\,k$$

```
> eqn5 := k = solve(eqn4, k);
```

$$eqn5 := k = \frac{1}{2}\ln\left(\frac{3}{2}\right)$$

```
> eqn6 := subs(eqn5, eqn1);
```

$$eqn6 := 120 = A\,\mathbf{e}^{(-\ln(3/2))}$$

```
> eqn7 := A = solve(eqn6, A);
```

$$eqn7 := A = 180$$

```
> formula_for_mass := subs({eqn5, eqn7}, basic_eqn);
```

$$formula\_for\_mass := m = 180\,\mathbf{e}^{(-1/2\ln(3/2)\,t)}$$

```
> answer := simplify(subs(t = 10, rhs(formula_for_mass)));
```

$$answer := \frac{640}{27}$$

5. The mass of an e. coli colony feasting in a nutrient broth doubles every 20 minutes.

How long does it take for the mass of the colony to triple?

Let T denote the tripling time in minutes

```
> restart;  interface(showassumed = 0): assume(T, positive);
> basic_eqn := m = A*2^(t/20);
```

$$basic\_eqn := m = A\,2^{\left(\frac{t}{20}\right)}$$

```
> tripling_eqn := 3*m = A*2^((t+T)/20);
```

$$tripling\_eqn := 3\,m = A\,2^{\left(\frac{t}{20}+\frac{T}{20}\right)}$$

```
> eqn1 := 3*basic_eqn;
```

$$eqn1 := 3\,m = 3\,A\,2^{\left(\frac{t}{20}\right)}$$

```
> eqn2 := rhs(tripling_eqn) = rhs(eqn1);
```

$$eqn2 := A\,2^{\left(\frac{t}{20}+\frac{T}{20}\right)} = 3\,A\,2^{\left(\frac{t}{20}\right)}$$

```
> eqn3 := map( z -> simplify(z/A), eqn2);
```

$$eqn3 := 2^{\left(\frac{t}{20}+\frac{T}{20}\right)} = 3\,2^{\left(\frac{t}{20}\right)}$$

```
> eqn4 := map( z -> simplify(z/2^(t/20)), eqn3);
```

$$eqn4 := 2^{\left(\frac{T}{20}\right)} = 3$$

```
> eqn5 := map( z -> simplify(ln(z)), eqn4);
```

$$eqn5 := \frac{1}{20}\,T\ln(2) = \ln(3)$$

```
> answer := T = solve(eqn5, T);
```

$$answer := T = \frac{20\ln(3)}{\ln(2)}$$

6. Calculate $\frac{d}{dx}\arcsin(\sqrt{x})$.

```
> general_formula := Diff(arcsin(u),u) = diff(arcsin(u),u);
```

$$general\_formula := \frac{d}{du}\arcsin(u) = \frac{1}{\sqrt{1-u^2}}$$

```
> Chain_Rule_form := Diff(arcsin(u(x)),x) = diff(arcsin(u(x)),x);
```

$$Chain\_Rule\_form := \frac{d}{dx}\arcsin(\mathrm{u}(x)) = \frac{\frac{d}{dx}\mathrm{u}(x)}{\sqrt{1-\mathrm{u}(x)^2}}$$

```
> given_problem := subs(u(x) = sqrt(x), Chain_Rule_form);
```

$$given\_problem := \frac{d}{dx}\arcsin(\sqrt{x}) = \frac{\frac{d}{dx}(\sqrt{x})}{\sqrt{1-x}}$$

```
> answer := lhs(given_problem) = rhs(given_problem);
```

$$answer := \frac{d}{dx}\arcsin(\sqrt{x}) = \frac{1}{2\sqrt{x}\sqrt{1-x}}$$

7. Calculate $\int_0^{\pi} x \sin(x)\, dx$.

```
> with(student):
> J := Int(x*sin(x),x);
```

$$J := \int x \sin(x)\, dx$$

```
> result_of_integration_by_parts := intparts(J,x);
```

$$result\_of\_integration\_by\_parts := -x\cos(x) - \int -\cos(x)\, dx$$

```
> antiderivative := value(result_of_integration_by_parts);
```

$$antiderivative := \sin(x) - x\cos(x)$$

```
> answer := subs(x=Pi,antiderivative) - subs(x=0,antiderivative);
```

$$answer := \sin(\pi) - \pi\cos(\pi) - \sin(0)$$

```
> simplify(answer);
```

$$\pi$$

8. Calculate $\int_0^1 x^2 e^x\, dx$.

```
> int(x^2*exp(x),x = 0 .. 1);
```

$$e - 2$$

```
>
> with(student):
```

```
> J := Int(x^2*exp(x),x);
```

$$J := \int x^2\, \mathrm{e}^x\, dx$$

```
> result_of_first_integration_by_parts := intparts(J,x^2);
  # u = x^2, dv = exp(x)*dx
```

$$result\_of\_first\_integration\_by\_parts := x^2\, \mathrm{e}^x - \int 2\, x\, \mathrm{e}^x\, dx$$

```
> result_of_second_integration_by_parts :=
      x^2*exp(x) - 2*intparts(Int(x*exp(x),x),x);
                # u = x, dv = exp(x)*dx
```

$$result\_of\_second\_integration\_by\_parts := x^2\, \mathrm{e}^x - 2\, x\, \mathrm{e}^x + 2 \int \mathrm{e}^x\, dx$$

```
> antiderivative := value(result_of_second_integration_by_parts);
```

$$antiderivative := x^2\, \mathrm{e}^x - 2\, x\, \mathrm{e}^x + 2\, \mathrm{e}^x$$

```
> answer := subs(x=1,antiderivative) - subs(x=0,antiderivative);
```

$$answer := \mathrm{e} - 2\, \mathrm{e}^0$$

```
> simplify(answer);
```

$$\mathrm{e} - 2$$

9. Calculate $\displaystyle \int_0^{\frac{\pi}{2}} \sin(x)^3 \cos(x)^{\left(\frac{1}{2}\right)}\, dx$ **.**

```
> eqn1 := Int(sin(x)^3*cos(x)^(1/2),x = 0 .. Pi/2) =
  Int((1-cos(x)^2)*cos(x)^(1/2)*sin(x),x = 0 .. Pi/2);
```

$$eqn1 := \int_0^{\frac{\pi}{2}} \sin(x)^3 \sqrt{\cos(x)}\, dx = \int_0^{\frac{\pi}{2}} (1 - \cos(x)^2) \sqrt{\cos(x)}\, \sin(x)\, dx$$

```
> eqn2 := lhs(eqn1) = changevar(u=cos(x),rhs(eqn1),u);
  # This command makes the change of variable u = cos(x)
```

$$eqn2 := \int_0^{\frac{\pi}{2}} \sin(x)^3 \sqrt{\cos(x)}\, dx = \int_0^1 (1 - u^2) \sqrt{u}\, du$$

```
> eqn3 := lhs(eqn2) = expand(rhs(eqn2));
```

$$eqn3 := \int_0^{\frac{\pi}{2}} \sin(x)^3 \sqrt{\cos(x)}\, dx = \int_0^1 \sqrt{u}\, du - \int_0^1 u^{(5/2)}\, du$$

```
> answer := lhs(eqn3) = value(rhs(eqn3));
```

$$answer := \int_0^{\frac{\pi}{2}} \sin(x)^3 \sqrt{\cos(x)}\, dx = \frac{8}{21}$$

10. Use the reduction formula

$$\int_0^1 x^n (1+x)^m\, dx = \frac{2^{(m+1)}}{m+1} - \frac{n}{m+1} \int_0^1 x^{(n-1)} (1+x)^{(m+1)}\, dx$$

and the equation $\int_0^1 (x+1)^6\, dx = \frac{127}{7}$ **to evaluate** $\int_0^1 x^2 (1+x)^4\, dx$**.**

```
> Integral := (n,m) -> Int(x^n*(1+x)^m, x = 0 .. 1);
```

$$Integral := (n, m) \rightarrow \int_0^1 x^n (1+x)^m\, dx$$

```
> reduction_formula := Integral(n,m) = 2^(m+1)/(m+1) -
  n/(m+1)*Integral(n-1,m+1);
```

$$reduction\_formula := \int_0^1 x^n (1+x)^m\, dx = \frac{2^{(m+1)}}{m+1} - \left(\frac{n}{m+1} \int_0^1 x^{(n-1)} (1+x)^{(m+1)}\, dx \right)$$

```
> eqn1 := subs({n=2,m=4}, reduction_formula);
```

$$eqn1 := \int_0^1 x^2 (1+x)^4\, dx = \frac{32}{5} - \frac{2}{5} \int_0^1 x (1+x)^5\, dx$$

```
> eqn2 := lhs(eqn1) = 32/5-2/5*subs({n=1,m=5},
  rhs(reduction_formula));
```

$$eqn2 := \int_0^1 x^2 (1+x)^4\, dx = \frac{32}{15} + \frac{1}{15} \int_0^1 (1+x)^6\, dx$$

```
> answer := lhs(eqn2) = 32/15 + 1/15*(127/7);
```

$$answer := \int_0^1 x^2 (1+x)^4 \, dx = \frac{117}{35}$$

11. Convert $\int_0^1 x^2 \sqrt{1+x^2} \, dx$ **to an integral of the form** $\int_a^b \sin(\theta)^p \cos(\theta)^q \, d\theta$.

A complete answer will have a, b, p, and q specified.

```
> restart; with(student):
> J := Int(x^2*f(x),x = 0 .. 1);
```

$$J := \int_0^1 x^2 \, f(x) \, dx$$

```
> J2 := changevar(x = tan(theta), J, theta);
```

$$J2 := \int_0^{\frac{\pi}{4}} \tan(\theta)^2 \, f(\tan(\theta)) \, (1 + \tan(\theta)^2) \, d\theta$$

```
> J3 := subs(f(tan(theta)) = (1+tan(theta)^2)^(1/2), J2);
```

$$J3 := \int_0^{\frac{\pi}{4}} \tan(\theta)^2 \, (1 + \tan(\theta)^2)^{(3/2)} \, d\theta$$

```
> J4 := subs(1+tan(theta)^2  = sec(theta)^2, J3);
```

$$J4 := \int_0^{\frac{\pi}{4}} \tan(\theta)^2 \, (\sec(\theta)^2)^{(3/2)} \, d\theta$$

```
> J5 := Int(tan(theta)^2*sec(theta)^3,theta = 0 .. 1/4*Pi);
```

$$J5 := \int_0^{\frac{\pi}{4}} \tan(\theta)^2 \sec(\theta)^3 \, d\theta$$

```
> J6 := subs({tan(theta) = sin(theta)/cos(theta), sec(theta) =
  1/cos(theta)}, J5);
```

$$J6 := \int_0^{\frac{\pi}{4}} \frac{\sin(\theta)^2}{\cos(\theta)^5}\, d\theta$$

12. Evaluate $\displaystyle\int \frac{x-2}{x\,(x-1)}\, dx$.

```
> eqn1 := Int((x-2)/x/(x-1),x) = Int(convert((x-2)/x/(x-1),
  parfrac, x),x);
```

$$eqn1 := \int \frac{x-2}{x\,(x-1)}\, dx = \int \frac{2}{x} - \frac{1}{x-1}\, dx$$

```
> eqn2 := lhs(eqn1) = value(rhs(eqn1))+C;
  #Maple works with complex numbers, so it does not insert
  absolute values
```

$$eqn2 := \int \frac{x-2}{x\,(x-1)}\, dx = 2\ln(x) - \ln(x-1) + C$$

13. Evaluate $\displaystyle\int \frac{5\,x^2 + x + 2}{(x+1)\,(x^2+1)}\, dx$.

```
> rational_fn := (5*x^2+x+2)/(x+1)/(x^2+1);
```

$$rational\_fn := \frac{5\,x^2 + x + 2}{(x+1)\,(1+x^2)}$$

```
> eqn1 := Int(rational_fn,x) = Int(convert(rational_fn, parfrac,
  x) ,x);
```

$$eqn1 := \int \frac{5\,x^2 + x + 2}{(x+1)\,(1+x^2)}\, dx = \int \frac{-1+2\,x}{1+x^2} + \frac{3}{x+1}\, dx$$

```
> lhs(eqn1) = value(rhs(eqn1))+C;
```

$$\int \frac{5x^2+x+2}{(x+1)(1+x^2)}\,dx = \ln(1+x^2) - \arctan(x) + 3\ln(x+1) + C$$

14. Evaluate $\displaystyle\int \frac{2x^2+3x+6}{x^2(x+2)}\,dx$

```
> rational_fn := (2*x^2+3*x+6)/(x^2)/(x+2);
```

$$rational\_fn := \frac{2x^2+3x+6}{x^2(x+2)}$$

```
> eqn1 := Int(rational_fn,x) = Int(convert(rational_fn, parfrac,
x) ,x);
```

$$eqn1 := \int \frac{2x^2+3x+6}{x^2(x+2)}\,dx = \int \frac{2}{x+2} + \frac{3}{x^2}\,dx$$

```
> lhs(eqn1) = value(rhs(eqn1))+C;
```

$$\int \frac{2x^2+3x+6}{x^2(x+2)}\,dx = 2\ln(x+2) - \frac{3}{x} + C$$